
Math 4530
Monday March 3

Computations related to the shape operator

     Here is a list of procedures to calculate the matrix of the shape operator, the principle curvatures, the 
mean curvature and the Gauss curvature, using a given patch X.  The procedures are modified from ones 
given on pages 119-121 of Oprea, using notation and linear algebra from class.  The procedures are 
illustrated with computations and pictures for the helicoid and the torus.
     This file lives on our Maple page, it is called surfacecurvatures.mws (also .pdf).

> restart:
with(linalg):
with(plots):

> assume(u,real); #this gets rid of that annoying "csgn" fcn
assume(v,real);

> #dot product
dp:=proc(X,Y)
X[1]*Y[1]+X[2]*Y[2]+X[3]*Y[3];
end:

> #2-norm, i.e. magnitude.
nrm:=proc(X)
sqrt(dp(X,X));
end:

> #cross product:
xp:=proc(X,Y)
local a,b,c;
a:=X[2]*Y[3]-X[3]*Y[2];
b:=X[3]*Y[1]-X[1]*Y[3];
c:=X[1]*Y[2]-X[2]*Y[1];
[a,b,c];
end:

> #Derivative matrix for mapping X:
DXq:=proc(X)
local Xu,Xv;
Xu:=matrix(3,1,[diff(X[1],u),diff(X[2],u),diff(X[3],u)]);
Xv:=matrix(3,1,[diff(X[1],v),diff(X[2],v),diff(X[3],v)]);
simplify(augment(Xu,Xv),radical,symbolic,trig);
end:

> #Matrix of first fundamental form:
gij:=proc(X)
local g11,g12,g22,Y;
Y:=evalm(DXq(X));
simplify(evalm(transpose(Y)&*Y),
   radical,symbolic,trig);
end:

> #unit normal:
U:=proc(X)



local Y,Z,s;
Y:=DXq(X);
Z:=xp(col(Y,1),col(Y,2));
s:=nrm(Z);
simplify(evalm((1/s)*Z),radical,symbolic,trig);
end:

> #matrix of second fundamental form:
hij:=proc(X)
local Y,Xu,Xv,Xuu,Xuv,Xvv,U1,h11,h12,h22;
Y:=DXq(X);
U1:=U(X);
Xu:=col(Y,1);
Xv:=col(Y,2);
Xuu:=[diff(Xu[1],u),diff(Xu[2],u),diff(Xu[3],u)];
Xuv:=[diff(Xu[1],v),diff(Xu[2],v),diff(Xu[3],v)];
Xvv:=[diff(Xv[1],v),diff(Xv[2],v),diff(Xv[3],v)];
h11:=dp(Xuu,U1);
h12:=dp(Xuv,U1);
h22:=dp(Xvv,U1);
simplify(matrix(2,2,[h11,h12,h12,h22]),
  radical,symbolic,trig);
end:

> #matrix of shape operator wrt basis {Xu,Xv}:
aij:=proc(X)
local Y,H,G;
H:=hij(X);
G:=gij(X);
simplify(evalm(inverse(G)&*H),
   radical,symbolic,trig);
end:

> #Gauss curvature
GK:=proc(X)
local A;
A:=aij(X);
simplify(det(A),radical,symbolic,trig);
end:

> #Mean curvature
MK:=proc(X)
local A;
A:=aij(X);
simplify(1/2*trace(A),radical,symbolic,trig);
end:

> #Principle curvatures and directions:
PK:=proc(X)
local Y;
Y:=aij(X);
eigenvects(Y);
end:



> test:=[u,v,u^2-v^2];

 := test [ ], ,u~ v~ −u~2 v~2

> DXq(test);

2 u~ −2 v~
> gij(test);
hij(test);
subs({u=0,v=0},aij(test));
subs({u=0,v=0},GK(test));
subs({u=0,v=0},MK(test));
subs({u=0,v=0},aij(test));

> torus:=[(2+cos(u))*cos(v),(2+cos(u))*sin(v),sin(u)];

 := torus [ ], ,( )+2 ( )cos u~ ( )cos v~ ( )+2 ( )cos u~ ( )sin v~ ( )sin u~
> gij(torus);

> hij(torus);

> aij(torus);

> GK(torus);

( )cos u~

+2 ( )cos u~
> MK(torus);

+1 ( )cos u~

+2 ( )cos u~
> aij(torus);

> plot3d(torus,u=0..2*Pi,v=0..2*Pi,color=GK(torus));

> hel:=[v*cos(u),v*sin(u),u];

 := hel [ ], ,v~ ( )cos u~ v~ ( )sin u~ u~
> gij(hel);
hij(hel);
aij(hel);
PK(hel);



MK(hel);
GK(hel);

> plot3d(hel,u=0..2*Pi,v=-3..3,color=GK(hel));

